Supplemental Analysis Example: Use of NOMCAR option in SAS v9.2 PROC SURVEYMEANS

May 9, 2011

This example demonstrates use of the NOMCAR (Not Missing Completely at Random) option in SAS V9.2 PROC SURVEYMEANS. The NOMCAR option treats those missing data on the outcome of interest as a separate domain and hence, performs an unconditional subpopulation analysis of non-missing cases. This ensures that despite missing data, the full structure of the complex sample is reflected in the estimates of statistics computed based on the non-missing data (after case-wise deletion).

This example uses a data set created in-stream (modified from an example from the SAS PROC SURVEYMEANS documentation) with missing data on spending on energy drinks. The use of the NOMCAR option is omitted in the first means analysis and the variance estimates are then compared to those from the second means analysis with use of the NOMCAR option. The comparison of the variance and standard errors show the expected increase in variance when the observed data is treated as a domain rather than a subset of the data set.

--------------------------------------------------------------------------------------------------

* example of use of NOMCAR option in surveymeans ;

data energydrink ;
  input Grade Spending @@;
  if Grade=7 then Prob=20/1824;
  if Grade=8 then Prob=9/1025;
  if Grade=9 then Prob=11/1151;
  Weight=1/Prob;
  datalines;
  7 7 7 7 8 . 9 10 7 . 7 10 7 3 8 20 8 19 7 2
  7 . 9 15 8 16 7 6 7 6 9 15 8 17 8 14 9 .
  9 8 9 7 7 3 7 12 7 4 9 14 8 18 9 9 7 2 7 1
  7 4 7 11 9 8 8 . 8 13 7 . 9 . 9 11 7 2 7 9
  7 . 7 4
  8 . 8 .
  9 . 9 . ;
run ;

data energydrink ;
set energydrink ;
  if spending=. then missinggroup=1 ; else missinggroup=0 ;
proc means n nmiss mean stderr ;
var spending ;
run ;

data StudentTotals;
  input Grade _total_;
  datalines;
  7 1824
  8 1025
  9 1151
; run ;
*If you think that those students who did not respond spend differently than those students who did respond, you can use the NOMCAR option to request the analysis to treat the respondents as a domain rather than exclude the nonrespondents.*

```
title 'Analysis of Energy Drink Spending with Missing Data Excluded from the Analysis';
proc surveymeans data=energydrink total=studenttotals mean stderr var ;
  strata Grade;
  var Spending;
  weight Weight;
run;
```

Analysis of Energy Drink Spending with Missing Data Excluded from the Analysis

The SURVEYMEANS Procedure

Data Summary

<p>| | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>Number of Strata</td>
<td>3</td>
<td></td>
</tr>
<tr>
<td>Number of Observations</td>
<td>46</td>
<td></td>
</tr>
<tr>
<td>Sum of Weights</td>
<td>4619.45051</td>
<td></td>
</tr>
</tbody>
</table>

Statistics

<table>
<thead>
<tr>
<th>Variable</th>
<th>Mean</th>
<th>Std Error of Mean</th>
<th>Var of Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>Spending</td>
<td>9.614865</td>
<td>0.527648</td>
<td>0.278413</td>
</tr>
</tbody>
</table>

```
title 'Analysis of Energy Drink Spending with NOMCAR option';
proc surveymeans data=energydrink total=StudentTotals nomcar mean stderr var ;
  strata Grade;
  var Spending;
  weight Weight;
run;
```

Analysis of Energy Drink Spending with NOMCAR option

The SURVEYMEANS Procedure

Data Summary

<p>| | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>Number of Strata</td>
<td>3</td>
<td></td>
</tr>
<tr>
<td>Number of Observations</td>
<td>46</td>
<td></td>
</tr>
<tr>
<td>Sum of Weights</td>
<td>4619.45051</td>
<td></td>
</tr>
</tbody>
</table>

Variance Estimation

<table>
<thead>
<tr>
<th>Method</th>
<th>Taylor Series</th>
</tr>
</thead>
<tbody>
<tr>
<td>Missing Values</td>
<td>NOMCAR</td>
</tr>
</tbody>
</table>

Statistics

<table>
<thead>
<tr>
<th>Variable</th>
<th>Mean</th>
<th>Std Error of Mean</th>
<th>Var of Mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>Spending</td>
<td>9.614865</td>
<td>0.689100</td>
<td>0.474859</td>
</tr>
</tbody>
</table>